

개념은 **쉽게**
기능은 **빠르게**
실무활용은 **바로**

실무
리더



회사에서 바로 **통**하는

김경자 지음

엑셀 매크로 & VBA

모든 버전 사용 가능

2007 2010 2013 2016 2019 Microsoft 365



무료 특별판

한빛미디어
Hanbit Media, Inc.



별책 부록 PDF

실무 프로그램 제작으로 VBA 프로그래밍 실전 감각 기르기

고급 실무 프로젝트 01 로그인 정보를 확인한 후 파일을 열어주는 보안 프로그램 만들기 - 003

▶ **회사통 키워드** 사용자 정의 폼, 레이블, 텍스트 상자, 명령 단추, Workbook_Open 이벤트, Workbook_BeforeClose 이벤트, For Next 반복문, For Each 반복문, If 조건문, Cells 속성, Visible 속성, Save 메서드, Show 메서드, Unload 문

STEP 01 시트 준비와 사용자 정의 폼 작성하기 006

STEP 02 로그인 정보를 확인하는 이벤트 프로시저 작성하기 009

STEP 03 통합 문서가 열리고 닫히기 직전에 프로그램이 실행되도록 이벤트 프로시저 작성하기 014

비법 노트 VBA 프로젝트 암호 설정하기 018

고급 실무 프로젝트 02 리모델링 수주 고객관리 프로그램 만들기 019

▶ **회사통 키워드** 사용자 정의 폼, 콤보 상자, 목록 상자, 텍스트 상자, 이미지, 확인란, 옵션 단추, 명령 단추, Initialize 이벤트, ListIndex 속성, SetFocus 메서드, SendKeys 메서드, GetOpenFilename 메서드, Show 메서드, Picture 속성, EntireRow 속성, Find 메서드, LoadPicture 함수, Date 함수, Unload 문, If 조건문, Do While 반복문

STEP 01 고객등록과 고객 검색/삭제 사용자 정의 폼 작성하기 023

STEP 02 고객등록 폼 이벤트 프로시저 작성하기 031

STEP 03 고객 검색/삭제 폼의 이벤트 프로시저 작성하기 045

실무 프로그램 제작으로 VBA 프로그래밍 실전 감각 기르기

VBA로 엑셀의 모든 기능을 코드로 구현할 수 있지만, 실제 업무에 사용하는 코드는 한정되어 있습니다. 따라서 실무에서 많이 사용되는 VBA 학습에서 하나의 완성된 프로그램을 제작하는 것은 앞으로 VBA를 활용해 업무에 적용하는 데 있어 중요한 경험이 됩니다. 특별 부록으로 제공되는 고급 실무 프로젝트를 통해 하나의 프로그램을 처음부터 끝까지 직접 따라 하며 제작해 VBA 프로그래밍 감각을 경험할 수 있습니다.

01

로그인 정보를 확인한 후 파일을 열어주는 보안 프로그램 만들기

실습 파일 | 별책부록/인사기록부보안파일.xlsx

완성 파일 | 별책부록/인사기록부보안파일(완성).xlsm

01 프로젝트 시작하기


엑셀에서 통합 문서 열기 암호를 설정하면 문서를 보호할 수 있지만 암호는 한 개만 사용할 수 있습니다. 여러 사람이 공동으로 사용하는 통합 문서라면 로그인 아이디와 비밀번호를 각각 다르게 사용하여 문서 열기 권한을 부여할 수 있도록 하면 편리합니다.

인사기록부 보안 파일을 열면 매크로가 실행되면서 아이디와 비밀번호를 확인하는 사용자 정의 폼이 나타나고, 이 폼에서 입력하는 아이디와 비밀번호를 통합 문서에 입력해 둔 정보와 비교한 후 일치할 경우 [인사기록부] 시트를 표시하는 프로그램을 만들어보겠습니다. 만약 입력하는 아이디와 비밀번호가 일치하지 않으면 보안 파일인 [인사기록부] 시트를 표시하지 않고 안내 메시지가 나타나도록 하겠습니다.

회사에서 바로 통하는 키워드


사용자 정의 폼, 레이블, 텍스트 상자, 명령 단추, Workbook_Open 이벤트, Workbook_BeforeClose 이벤트, For Next 반복문, For Each 반복문, If 조건문, Cells 속성, Visible 속성, Save 메서드, Show 메서드, Unload 문

02 프로젝트 예제 미리 보기



임직원 인사기록부										
사번	성명	소속	직급	인사고과	근태점수	평가	자택주소	우편번호	휴대폰번호	
MDS212320	강규영	CS경영팀	사원	58	6	D	서울 특별시 구로구 구로동 197-12	44659	010-9546-662	
MDS131661	강균성	인사팀	차장	68	6	C	서울 관악구 신림동 56-1	15101	010-2478-157	
MDS114233	강미경	CS경영팀	부장	75	7	B	서울시 강남구 논현동 18402	47284	010-3424-752	
MDS130320	강민도	미래전략연구실	차장	69	6	B	서울시 송파구 풍납1동 123-45	13075	010-7146-356	
MDS136526	강희선	재무회계팀	부장	65	8	C	서울 송파구 잠실1동 10	13822	010-8839-389	
MDS109714	고경희	미래전략연구실	과장	84	8	A	서울시 성북구 화랑로11길 26, 102-1	15783	010-8449-559	
MDS141258	고미선	기획조정실	과장	61	6	C	서울 중구 다동 103 등부다동빌딩	10018	010-9809-932	
MDS216749	고찬근	인사팀	차장	85	6	A	서울시 금천구 시흥1동 920	12185	010-5258-694	
MDS040362	구선욱	인사팀	대리	74	7	B	서울시 은평구 용암1동 3-777	15302	010-1702-176	
MDS133903	구유허	미래전략연구실	차장	85	7	A	서울 강남구 역삼동 642-6	13508	010-2792-266	
MDS135258	국민영	품질보증부	사원	65	6	C	경기도 군포시 당동 730	43501	010-0025-694	
MDS140258	권관영	인사팀	사원	62	6	C	서울시 강서구 가양 1487	15720	010-0459-159	
MDS216545	권미경	CS경영팀	대리	51	5	D	서울시 서대문구 홍은동 8312	10086	010-9276-433	
MDS129521	권민정	경영감사팀	사원	69	6	B	경기도 용인시 수지구 성북2로	15189	010-2050-700	


한눈에
보는
작업순서



03 핵심기능 미리 보기


STEP 01 시트 준비와 사용자 정의 폼 작성하기

- 1 통합 문서가 열렸을 때 기본으로 보여줄 시트와 로그인 정보를 확인한 후에 보여줄 보안 문서 시트를 준비합니다.
- 2 로그인 정보로 아이디와 비밀번호를 입력할 수 있는 사용자 정의 폼을 삽입하고, 레이블과 텍스트 상자, 명령 단추를 폼에 추가합니다.




STEP 02 로그인 정보를 확인하는 이벤트 프로시저 작성하기

- 1 로그인 폼에서 [로그인] 단추를 클릭하면 실행되는 Click 이벤트 프로시저를 추가합니다.
- 2 필요한 변수를 선언하고 변수에 초깃값을 대입합니다.
- 3 로그인 폼에서 입력한 아이디와 비밀번호가 [로그인정보] 시트의 데이터와 일치하는지 확인하는 조건문과 반복문을 입력합니다.
- 4 정보가 일치하면 [인사기록부] 시트와 [로그인정보] 시트를 숨기기 취소하고, [시작화면] 시트는 숨기는 코드를 입력합니다.
- 5 정보가 일치하지 않으면 메시지 대화상자를 표시하고 폼을 닫는 코드를 입력합니다.



STEP 03 통합 문서가 열리고 닫히기 직전에 프로그램이 실행되도록 이벤트 프로시저 작성하기

- 1 통합 문서가 열릴 때 폼이 실행되도록 'Workbook_Open' 이벤트 프로시저를 작성합니다.
- 2 통합 문서가 닫히기 직전에 [인사기록부] 시트와 [로그인정보] 시트를 숨기고, [시작화면] 시트만 표시하도록 'Workbook_BeforeClose' 이벤트 프로시저를 작성합니다.



STEP 01

시트 준비와 사용자 정의 폼 작성하기

이 통합 문서에는 매크로가 실행되지 않은 상태에서 보여줄 시트와 로그인 정보를 확인한 후 보여줄 시트가 구분되어 있습니다. 로그인 정보를 확인하는 프로그램을 만들었다 하더라도 매크로 보안 경고에서 [콘텐츠 사용]을 클릭하지 않으면 프로그램이 의미가 없기 때문에 매크로가 제외한 상태에서 문서를 열었을 때는 중요한 시트를 볼 수 없도록 숨겨두기 위해서입니다. 준비된 시트를 확인하고 로그인 정보를 입력하는 사용자 정의 폼을 작성합니다.


01 시트 확인하기 실습 파일을 열고 ❶ [시작화면] 시트 탭을 클릭합니다. 이 시트는 매크로가 실행되지 않았을 때 기본으로 보여줄 시트로, 매크로가 실행되고 로그인 정보가 확인되면 숨겨집니다. ❷ [인사기록부] 시트 탭을 클릭합니다. 이 시트는 로그인 정보가 확인된 후 보여줄 시트로, 파일이 닫힐 때는 VBA 코드에 의해 다시 숨겨집니다.

The screenshot shows the Excel interface with the '시작화면' (Start Screen) sheet selected. The sheet contains a login form with a red lock icon and a message: '[문서 로그인 안내] 인사기록부는 보안 파일입니다. 내용을 확인하려면 [콘텐츠 사용]을 허용 한 후 로그인 정보를 입력하세요.' The '시작화면' tab is highlighted with a red circle and the number 1. Below it, the '인사기록부' (Employee Record Book) sheet is visible, showing a table of employee information. The '인사기록부' tab is also highlighted with a red circle and the number 2.


사번	성명	소속	직급	인사고과	근태점수	평가	자택주소	우편번호	휴대폰번호
MDS212320	강규영	CS경영팀	사원	58	6	D	서울특별시 구로구 구로동 197-12	44659	010-9546-6622
MDS131661	강균성	인사팀	차장	68	6	C	서울 관악구 신림동 56-1	15101	010-2478-157
MDS114233	강미경	CS경영팀	부장	75	7	B	서울시 강남구 논현동 18402	47284	010-3424-7522
MDS130320	강민도	미래전략연구실	차장	69	6	B	서울시 송파구 풍납1동 123-45	13075	010-7146-3562
MDS136526	강희선	재무회계팀	부장	65	8	C	서울 송파구 잠실1동 10	13822	010-8839-3892
MDS109714	고경희	미래전략연구실	과장	84	8	A	서울시 성북구 화랑로11길 26, 102-1	15783	010-8449-559
MDS141258	고미선	기획조정실	과장	61	6	C	서울 중구 다동 103 동부다동빌딩	10018	010-9809-9322
MDS216749	고찬근	인사팀	차장	85	6	A	서울시 금천구 시흥1동 920	12185	010-5258-6944
MDS040362	구선욱	인사팀	대리	74	7	B	서울시 은평구 용암1동 3-777	15302	010-1702-1762
MDS133903	구유정	미래전략연구실	차장	85	7	A	서울 강남구 역삼동 642-6	13508	010-2792-2662
MDS135258	국민영	품질보증부	사원	65	6	C	경기도 군포시 당동 730	43501	010-0025-6942
MDS140258	권관영	인사팀	사원	62	6	C	서울시 강서구 가양 1487	15720	010-0459-159
MDS216545	권미경	CS경영팀	대리	51	5	D	서울시 서대문구 홍은동 8312	10086	010-9276-4332
MDS2129521	권민정	인사팀	사원	69	6	B	경기도 용인시 수지구 성북2로	15189	010-2050-7002

실력향상 시트를 숨길 때 Visible 속성값을 **xISheetVeryHidden**으로 설정하면 시트 탭을 마우스 오른쪽 버튼으로 클릭하여 [숨기기 취소]를 클릭하는 작업으로는 숨기기 취소를 할 수 없습니다. VBA 코드로 Visible 속성을 다시 **xISheetVisible**로 변경해야 시트를 표시할 수 있습니다.

02 [로그인정보] 시트 탭을 클릭합니다. 이 시트에는 로그인 가능한 아이디와 비밀번호 목록이 입력되어 있고 새로운 사용자가 있거나 로그인 정보를 변경할 때는 이 시트에서 직접 수정할 수 있습니다. 이 시트도 로그인 정보가 확인된 후 보여줄 시트이며 파일이 닫힐 때는 VBA 코드에 의해 다시 숨겨집니다.





03 사용자 정의 폼 삽입하기 ❶ [Visual Basic 편집기]에서 [삽입]-[사용자 정의 폼]을 클릭합니다. ❷ [속성] 창에서 [이름]에 **frm로그인**을, ❸ [Caption]에 **사용자 확인**을 입력합니다.





실력향상 속성 창이 표시되지 않으면 [보기]-[속성 창]을 클릭합니다.

04 폼에 컨트롤 추가하기 ❶ 폼 오른쪽 하단 모서리의 크기 조절점을 드래그하여 폼 크기를 변경합니다. ❷ [도구 상자]에서 [레이블A], [텍스트 상자B], [명령 단추C]를 각각 클릭하여 폼에 추가한 후 각 컨트롤의 속성을 다음과 같이 변경합니다.



번호	컨트롤 종류	속성 변경
❶	레이블	• Caption : 아이디, 비밀번호
❷	텍스트 상자	• 이름 : txtID
❸	텍스트 상자	• 이름 : txtPW • PasswordChar : *
❹	명령 단추	• 이름 : cmd로그인 • Caption : 로그인
❺	명령 단추	• 이름 : cmd취소 • Caption : 취소

05 폼 실행하기 ❶ [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행]을 클릭합니다. 폼이 실행됩니다. ❷ [닫기X]를 클릭하여 폼을 닫습니다.




STEP 02


로그인 정보를 확인하는 이벤트 프로시저 작성하기

로그인 폼에서 아이디와 비밀번호를 입력한 후 [로그인]을 클릭하면 [로그인정보] 시트에 입력된 데이터와 일치하는지 확인하는 이벤트 프로시저를 작성해보겠습니다. 로그인 정보가 일치할 경우 [인사기록부] 시트와 [로그인정보] 시트를 화면에 표시하고, [시작화면] 시트는 숨기기 취소합니다. 만약 로그인 정보가 일치하지 않으면 메시지 대화상자를 표시하고 폼을 닫습니다.

06 로그인 단추의 Click 이벤트 프로시저 작성하기 [frm로그인] 폼에 추가한 [로그인] 명령 단추를 더블클릭합니다. [코드] 창에 'cmd로그인_Click' 이벤트 프로시저가 추가됩니다.



07 폼에 입력하는 아이디와 비밀번호가 [로그인정보] 시트의 내용과 일치하는지 확인하고 일치하면 [인사기록부] 시트와 [로그인정보] 시트를 보여주는 조건문과 반복문을 다음과 같이 입력합니다.




```


1 Private Sub cmd로그인_Click()
2 Dim 행번호 As Integer
3 Dim 로그인시트 As Worksheet
4 Dim 시트 As Worksheet
5 Set 로그인시트 = Sheets("로그인정보")
6 For 행번호 = 2 To 로그인시트.Range("A1").End(xlDown).Row
7     If 로그인시트.Cells(행번호, 1).Value = Me.txtID.Value And _
        로그인시트.Cells(행번호, 2).Value = Val(Me.txtPW.Value) Then
8         MsgBox "아이디와 비밀번호가 확인되었습니다"
9         For Each 시트 In Worksheets
10             If 시트.Name <> "시작화면" Then 시트.Visible = xlSheetVisible
11         Next
12         Sheets("시작화면").Visible = xlSheetHidden
13         Sheets("인사기록부").Select
14         Unload Me
15         Exit Sub
16     End If
17 Next
18 MsgBox "로그인 정보가 일치하지 않습니다"
19 Unload Me
End Sub

```



- 1 [로그인] 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저입니다.
- 2 로그인 정보를 확인할 때 사용할 반복문 카운터 변수를 선언합니다.
- 3 [로그인정보] 시트를 저장할 '로그인시트' 변수를 선언합니다.
- 4 현재 통합 문서의 모든 시트를 저장하여 반복할 때 사용할 '시트' 변수를 선언합니다.
- 5 '로그인시트' 변수에 [로그인정보] 시트를 대입합니다.
- 6 반복문을 시작하여 [로그인정보] 시트의 A열에 입력된 셀 데이터 개수만큼 7 ~ 16을 반복 실행합니다.
반복문의 시작값은 2이고, 최종값은 로그인시트.Range("A1").End(xlDown).Row에 의해 [로그인] 시트의 [A1] 셀에서 아래쪽으로 데이터가 입력된 범위의 행 개수를 계산하여 결정합니다.
- 7 정보가 일치하는지 확인하는 조건문을 시작합니다. 한 문장인데 줄 언더바(_)를 사용하여 두 줄로 입력했습니다. [로그인시트]의 A열 데이터와 폼의 아이디 입력란의 값이 일치하고 B열 데이터와 폼의 [비밀번호] 입력란의 값이 일치하는지 확인합니다. 텍스트 상자에 입력되는 값은 문자로 인식되므로 Val 함수를 사용하여 숫자 형식으로 변환한 후 비교합니다. 조건식이 참이면 8 ~ 15를 실행하고 거짓이면 17에서 Next를 만나 6으로 이동하여 반복횟수가 남아 있으면 다음 정보를 다시 비교합니다.

- ⑧ 로그인 정보가 확인되었다는 메시지 대화상자를 나타냅니다.
- ⑨ ~ ⑪ [시작화면] 시트를 제외한 나머지 시트를 모두 표시하는 반복문입니다. 현재 통합 문서의 모든 워크시트를 '시트' 변수에 대입한 후 시트 이름이 '시작화면'이 아니면 숨기기 취소합니다. If 조건문은 한 줄로 입력하여 End If가 없습니다.
- ⑫ [시작화면] 시트를 숨기기합니다. Visible의 속성값으로 **xlSheetHidden**을 사용하면 시트 탭을 마우스 오른쪽 버튼으로 클릭하여 [숨기기 취소]를 할 수 있습니다.
- ⑬ [인사기록부] 시트를 선택합니다.
- ⑭ 'frm로그인' 폼을 닫습니다.
- ⑮ 프로그램을 종료합니다. 로그인 정보가 확인되면 ⑩ ~ ⑪를 실행하지 않아야 하므로 필요한 문장입니다.
- ⑯ If 문을 종료합니다.
- ⑰ 반복문을 종료합니다.
- ⑱ 로그인 정보가 일치하지 않는다는 메시지 대화상자를 나타냅니다. 로그인 정보가 일치하지 않은 상태에서 반복문이 종료된 후에 실행되는 문장입니다.
- ⑲ 'frm로그인' 폼을 닫습니다.

08 취소 단추의 Click 이벤트 프로시저 작성하기 ① [개체]에서 [cmd취소]를 클릭합니다. 'cmd취소_Click' 이벤트 프로시저가 추가됩니다. [취소]를 클릭했을 때 현재 폼이 닫히면 되므로 ② **Unload Me**를 입력합니다.




09 결과 확인하기 ① [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행]을 클릭합니다. [사용자 확인] 대화상자의 ② [아이디]에 onwings, [비밀번호]에 1234를 입력하고 ③ [로그인]을 클릭합니다. 로그인 정보가 확인되었다는 메시지가 나타나면 ④ [확인]을 클릭합니다. [시작화면] 시트는 숨겨지고, [인사기록부] 시트와 [로그인정보] 시트만 표시됩니다.



사번	성명	소속	직급	인사고과	근태점수	평가	자택주소	우편번호	휴대폰번호
MDS212320	강규영	CS경영팀	사원	58	6	D	서울 특별시 구로구 구로동 197-12	44659	010-9546-662
MDS131661	강균성	인사팀	차장	68	6	C	서울 관악구 신림동 56-1	15101	010-2478-157
MDS114233	강미경	CS경영팀	부장	75	7	B	서울시 강남구 논현동 18402	47284	010-3424-752
MDS130320	강민도	미래전략연구실	차장	69	6	B	서울시 송파구 풍납1동 123-45	13075	010-7146-356
MDS136526	강희선	재무회계팀	부장	65	8	C	서울 송파구 잠실1동 10	13822	010-8839-389
MDS109714	고경희	미래전략연구실	과장	84	8	A	서울시 성북구 화랑로11길 26, 102-1	15783	010-8449-559
MDS141258	고미선	기획조정실	과장	61	6	C	서울 중구 다동 103 동부다동빌딩	10018	010-9809-932
MDS216749	고찬근	인사팀	차장	85	6	A	서울시 금천구 시흥1동 920	12185	010-5258-694
MDS040362	구선옥	인사팀	대리	74	7	B	서울시 은평구 용암1동 3-777	15302	010-1702-176
MDS133903	구유정	미래전략연구실	차장	85	7	A	서울 강남구 역삼동 642-6	13508	010-2792-266
MDS135258	국민영	품질보증부	사원	65	6	C	경기도 군포시 당동 730	43501	010-0025-694
MDS140258	권관영	인사팀	사원	62	6	C	서울시 강서구 가양 1487	15720	010-0459-159
MDS216545	권미경	CS경영팀	대리	51	5	D	서울시 서대문구 홍은동 8312	10086	010-9276-433
MDS129521	권민정	경영감사팀	사원	69	6	B	경기도 용인시 수지구 성북2로	15189	010-2050-700

10 [Visual Basic 편집기]의 [표준] 도구 모음에서 ❶ [Sub/사용자 정의 폼 실행]을 클릭합니다. [사용자 확인] 대화상자에 ❷ 아이디와 비밀번호를 임의로 입력한 후 ❸ [로그인]을 클릭합니다. 로그인 정보가 일치하지 않는다는 메시지가 나타납니다. ❹ [확인]을 클릭합니다.




실력향상 로그인 정보가 일치하지 않으면 [시작화면] 시트만 표시되고 다른 시트들은 모두 숨기겨져야 하는데 이 부분은 'Workbook_BeforeClose' 이벤트 프로시저에서 해결합니다.

STEP 03


통합 문서가 열리고 닫히기 직전에 프로그램이 실행되도록 이벤트 프로시저 작성하기

통합 문서가 열릴 때 폼이 실행되도록 'Workbook_Open' 이벤트 프로시저를 작성하고, 통합 문서가 닫히기 직전에 [인사기록부] 시트와 [로그인정보] 시트를 숨기고 [시작화면] 시트만 표시하도록 'Workbook_BeforClose' 이벤트 프로시저를 작성해보겠습니다.


11 현재 통합 문서의 Open 이벤트 프로시저 작성하기 [Visual Basic 편집기]의 [프로젝트] 탐색기 창에서 ❶ [현재_통합_문서]를 더블클릭합니다. [현재_통합_문서 (코드)] 창이 표시되면 ❷ [개체]에서 [Work book]을 클릭합니다. 'Workbook_Open' 이벤트 프로시저가 추가됩니다.




12 파일이 열릴 때 폼이 실행되도록 frm로그인.Show를 입력합니다.



13 현재 통합 문서의 BeforClose 이벤트 프로시저 작성하기 [프로시저]에서 [BeforClose]를 클릭합니다. 'Workbook_BeforClose' 이벤트 프로시저가 추가됩니다.



14 파일이 닫히기 전에 [시작화면] 시트를 다시 표시하고 다른 시트는 모두 숨기기는 코드를 입력합니다. 이때 시트를 숨기기 위해 사용하는 Visible 속성값으로 **xlSheetVeryHidden**을 사용해야 시트 탭을 마우스 오른쪽 버튼으로 클릭하여 [숨기기 취소]를 실행할 수 없습니다.




- ① Private Sub Workbook_BeforClose(Cancel As Boolean)
- ② Dim 시트 As Worksheet
- ③ Sheets("시작화면").Visible = xlSheetVisible
- ④ For Each 시트 In Worksheets
- ⑤ If 시트.Name <> "시작화면" Then 시트.Visible = xlSheetVeryHidden
- ⑥ Next
- ⑦ ThisWorkbook.Save
- ⑧ End Sub


- 1 통합 문서가 닫히기 직전에 실행되는 이벤트 프로시저입니다.
- 2 반복문에 사용할 '시트' 변수를 선언합니다.
- 3 [시작화면] 시트를 숨기기 취소합니다. 다른 시트를 숨기기 전에 최소 한 개의 시트는 숨기기 취소가 되어 있어야 하므로 [시작화면] 시트의 숨기기 취소가 먼저 실행되도록 합니다.
- 4 반복문을 시작합니다. 현재 통합 문서의 모든 시트를 '시트' 변수에 저장한 후 시트 수만큼 5를 반복 실행합니다.
- 5 시트 이름이 '시작화면'이 아니면 숨기기 취소하는 조건문입니다. Visible 속성값을 xlSheetVeryHidden으로 설정하면 워크시트에서 숨기기 취소를 할 수 없습니다. 한 줄로 입력한 If 조건문으로, End If가 없습니다.
- 6 반복문을 종료합니다.
- 7 현재 통합 문서를 저장합니다.

실력향상 워크시트 개체의 Visible 속성에 사용되는 상수에는 xlSheetVisible, xlSheetHidden, xlSheetVeryHidden이 있습니다. xlSheetVisible은 시트를 표시합니다. xlSheetHidden과 xlSheetVeryHidden은 시트를 숨기기하는데, xlSheetHidden은 워크시트에서 숨기기 취소를 할 수 있고, xlSheetVeryHidden은 숨기기 취소를 할 수 없습니다. xlSheetVeryHidden으로 지정하면 시트 탭을 마우스 오른쪽 버튼으로 클릭했을 때 [숨기기 취소] 목록에 시트 이름이 표시되지 않습니다. VBA로 다시 Visible 속성을 xlSheetVisible로 설정해야 화면에 표시됩니다.

15 결과 확인하기 1 [파일] 탭-[다른 이름으로 저장]을 클릭하여 2 파일 형식에서 [Excel 매크로 사용 통합 문서(*.xlsm)]을 클릭한 후 3 [저장]을 클릭합니다. 저장한 파일을 닫고 다시 열면 매크로가 실행되지 않아 [시작화면] 시트만 표시됩니다. 4 [콘텐츠 사용]을 클릭합니다.





16 ①[아이디]에 **onwings**, [비밀번호]에 **1234**를 입력하고 ②[로그인]을 클릭합니다. 로그인 정보가 확인되었다는 메시지가 나타나면 ③[확인]을 클릭합니다. [시작화면] 시트는 숨기기되고, [인사기록부] 시트와 [로그인정보] 시트만 표시됩니다.



사번	성명	소속	직급	인사고과	근태점수	평가	자택주소	우편
MD5212320	강규영	CS경영팀	사원	58	6	D	서울특별시 구로구 구로동 197-12	444
MD5131661	강금성	인사팀	차장	68	6	C	서울 관악구 신림동 56-1	151
MD5114233	강미경	CS경영팀	부장	75	7	B	서울시 강남구 논현동 18402	472
MD5130320	강민도	미래전략연구실	차장	69	6	B	서울시 송파구 풍납1동 123-45	136
MD5136526	강희선	재무회계팀	부장	65	8	C	서울 송파구 잠실1동 10	138
MD5109714	고경희	미래전략연구실	과장	84	8	A	서울시 성북구 화랑로11길 26, 102-1	157
MD5141258	고미선	기획조정실	과장	61	6	C	서울 중구 다동 103 동부다동빌딩	100
MD5216749	고찬근	인사팀	차장	85	6	A	서울시 금천구 시흥1동 920	122
MD5040362	구인욱	인사팀	대리	74	7	B	서울시 은평구 은평1동 3-777	151
MD5133903	구유정	미래전략연구실	차장	85	7	A	서울 강남구 역삼동 642-6	135
MD5135258	국민영	품질보증부	사원	65	6	C	경기도 군포시 당동 730	431
MD5140258	권관영	인사팀	사원	62	6	C	서울시 강서구 가양 1487	157
MD5216545	권미경	CS경영팀	대리	51	5	D	서울시 서대문구 홍은동 8312	100
MD5129521	권민정	경영감사팀	사원	69	6	B	경기도 용인시 수지구 성북2로	151

17 파일을 닫고 다시 엽니다. ① 이번에는 아이디와 비밀번호를 임의로 입력한 후 ②[로그인]을 클릭합니다. 로그인 정보가 일치하지 않는다는 메시지가 나타나면 ③[확인]을 클릭합니다. [시작화면] 시트만 표시됩니다.




실력향상 다른 이름으로 저장한 매크로 통합 문서를 처음 열었을 때는 보안 경고가 표시되지만 같은 파일을 다시 열면 신뢰할 수 있는 문서로 등록되어 보안 경고 메시지가 표시되지 않습니다.




VBA 프로젝트 암호 설정하기

[Visual Basic 편집기]에서 작성한 사용자 정의 폼이나 모듈을 다른 사람이 확인하거나 변경할 수 없도록 VBA 프로젝트 암호를 설정할 수 있습니다.


01 [Visual Basic 편집기]에서 [도구] - [VBAProject 속성]을 클릭합니다.



02 [프로젝트 속성] 대화상자의 ①[보호] 탭을 클릭하고 [프로젝트 잠금]에서 ②[읽기 전용으로 프로젝트 잠금]에 체크합니다. [프로젝트 속성 보기 암호]에서 ③[암호]와 [암호 확인]을 동일하게 입력한 후 ④[확인]을 클릭합니다.



03 VBA 프로젝트가 보호되었는지 확인하기 위해 현재 통합 문서를 닫은 후 다시 엽니다. [Visual Basic 편집기]의 [프로젝트] 탐색기 창에서 파일을 클릭하면 암호를 입력하는 대화상자가 나타납니다.



리모델링 수주 고객관리 프로그램 만들기

실습 파일 | 별책부록/리모델링고객관리.xlsx

완성 파일 | 별책부록/리모델링고객관리(완성).xlsm

01 프로젝트 시작하기


주택, 아파트의 리모델링을 의뢰하는 고객의 정보를 등록하고, 수정 또는 삭제할 수 있는 고객관리 프로그램을 만들어보겠습니다.

사용자 정의 폼을 이용하여 [고객등록] 폼과 [고객 검색/삭제] 폼을 삽입한 후 [고객등록] 폼에서는 신규 고객의 정보와 리모델링 항목, 비용, 사진 등을 등록하여 [고객DB] 시트에 자동 입력되도록 합니다. [고객 검색/삭제] 폼에서는 검색 단어를 입력하면 [고객DB] 시트에서 데이터를 검색하여 폼에 표시하고, 수정사항이 있을 경우 폼에서 수정하면 [고객DB] 시트에도 자동 수정되도록 프로그램을 작성해보겠습니다.

회사에서 바로 통하는 키워드

사용자 정의 폼, 콤보 상자, 목록 상자, 텍스트 상자, 이미지, 확인란, 옵션 단추, 명령 단추, Initialize 이벤트, ListIndex 속성, SetFocus 메서드, SendKeys 메서드, GetOpenFilename 메서드, Show 메서드, Picture 속성, EntireRow 속성, Find 메서드, LoadPicture 함수, Date 함수, Unload 문, If 조건문, Do While 반복문

02 프로젝트 예제 미리 보기



고객등록

기본사항
 주택유형 계약일자 2020-03-20 공사일자 2020-04-01

고객정보
 성명 김길자 휴대폰번호 010-1111-2222 (+하이픈 포함 입력)
 주소 서울시 강남구 역삼동 21-2
 email kji@daum.net
 특기사항 신규 고객
 고객사진 사진 찾기
 김희숙.jpg

리모델링 항목
☒ 기실 ☒ 욕실
☒ 주방 ☐ 현관
☒ 안방 ☐ 발코니
☒ 작음방

금액
 공사비 4500000
 결제방법 ☒ 현금 ☐ 카드

신규 등록 닫기 입력 초기화 고객DB 조회 고객검색

고객 검색/삭제


검색성명 휴대폰번호 검색하기
 (1글자 이상 입력) (숫자 4글자 입력)

성명	주택유형	휴대폰번호
<input type="radio"/> 김경자	아파트	010-5726-3626
<input type="radio"/> 김태근	아파트	010-4793-2693
<input checked="" type="radio"/> 김한수	연립주택	010-0063-1111
<input type="radio"/> 여환구	다가구주택	010-6663-5148
<input type="radio"/> 도주철	단독주택	010-1424-8048
<input type="radio"/> 한보라	단독주택	010-6802-5599
<input type="radio"/> 김영미	아파트	010-7965-7786
<input type="radio"/> 박연미	아파트	010-3784-5584
<input type="radio"/> 김태준	연립주택	010-7521-7276
<input type="radio"/> 김희숙	다가구주택	010-2419-4499
<input type="radio"/> 윤용상	아파트	010-9523-6838
<input type="radio"/> 박재득	아파트	010-8879-4364
<input type="radio"/> 이광선	아파트	010-6147-8954
<input type="radio"/> 박정아	아파트	010-4057-5160
<input type="radio"/> 안재성	아파트	010-1702-1764
<input type="radio"/> 김경자	아파트	010-1111-2222

고객상세정보
 성명 김한수 리모델링 항목
 주택유형 연립주택 주방 작음방
 계약일자 2020-03-08
 공사일자 2020-03-28 사진
 휴대폰번호 010-0063-1111
 공사비 6728000
 결제방법 카드
 주소 서울시 종로구 함경궁로 명동2가 명신A 13-
 email hongnana@daum.net
 특기사항
 수정사항 등록 고객 삭제

신규 등록 고객 DB 닫기


한눈에 보는 작업순서



03 핵심기능 미리 보기


STEP 01 고객등록과 검색 사용자 정의 폼 작성하기

- 1 사용자 정의 폼을 두 개 삽입한 후 컨트롤들을 추가하여 [고객등록] 폼을 작성합니다.
- 2 [고객 검색/삭제] 폼을 작성합니다.
- 3 각 폼의 탭 순서를 지정합니다.



STEP 02 고객등록 폼 이벤트 프로시저 작성하기

- 1 'Initialize' 이벤트 프로시저를 추가하여 콤보 상자의 목록을 표시하고 'txt계약일자' 텍스트 상자에 오는 날짜를 표시합니다.
- 2 [사진 찾기] 단추의 'Click' 이벤트 프로시저를 추가하여 고객 사진을 선택하면 이미지 컨트롤에 표시하고 사진 파일 이름을 'txt사진파일명' 상자에 표시합니다.
- 3 [신규 등록] 단추의 'Click' 이벤트 프로시저를 추가하여 폼에 입력한 내용이 [고객DB] 시트에 자동 입력되도록 합니다.
- 4 [입력 초기화] 단추의 'Click' 이벤트 프로시저를 추가하여 폼의 컨트롤 값이 모두 초기화되도록 합니다.



STEP 03 고객 검색/삭제 폼 이벤트 프로시저 작성하기

- 1 [검색하기]의 'Click' 이벤트 프로시저를 추가하여 성명과 전화번호로 [고객DB] 시트에서 고객 정보를 검색하여 '1st검색' 목록 상자에 표시합니다.
- 2 '1st검색' 목록 상자의 'Click' 이벤트 프로시저를 추가하여 선택한 고객의 상세 정보가 [고객 검색/삭제] 폼에 표시되도록 합니다.
- 3 [수정 사항 등록]의 'Click' 이벤트 프로시저를 추가하여 수정한 고객 정보가 [고객DB] 시트에 입력되도록 합니다.
- 4 [고객 삭제]의 'Click' 이벤트 프로시저를 추가하여 선택한 고객 정보를 [고객DB] 시트에서 삭제합니다.
- 5 [고객등록] 폼과 [고객 검색/삭제] 폼을 실행하는 프로시저를 공용 모듈에 작성하고 [메뉴] 시트에 단추(양식 컨트롤)를 추가하여 프로시저를 연결합니다.

The screenshot displays the VBA development environment and the application interface. The top part shows the VBA editor with the following code for the 'Click' event of the '검색하기' button:

```
Private Sub cmd추정_Click()
Call 고객정보검색

셀워치, Offset(0, 4) = Me.txt조회성명
셀워치, Offset(0, 1) = Me.txt조회전화
셀워치, Offset(0, 2) = Me.txt조회계좌
셀워치, Offset(0, 3) = Me.txt조회사원
셀워치, Offset(0, 5) = Me.txt조회주소
셀워치, Offset(0, 6) = Me.txt조회주소
셀워치, Offset(0, 7) = Me.txt조회이메일
셀워치, Offset(0, 8) = Me.txt조회특기
셀워치, Offset(0, 17) = Val(Me.txt조회공시비)
셀워치, Offset(0, 18) = Me.txt조회결제방법

Call Cmd검색_Click
MsgBox Me.txt조회성명 & "님의 정보가 수정되었습니다"
End Sub

Private Sub lst검색_Click()
Call 고객정보검색

Me.txt조회성명 = 셀워치.Offset(0, 4)
Me.txt조회전화 = 셀워치.Offset(0, 1)
Me.txt조회계좌 = 셀워치.Offset(0, 2)
Me.txt조회사원 = 셀워치.Offset(0, 3)
Me.txt조회주소 = 셀워치.Offset(0, 5)
Me.txt조회주소 = 셀워치.Offset(0, 6)
Me.txt조회이메일 = 셀워치.Offset(0, 7)
Me.txt조회특기 = 셀워치.Offset(0, 8)
```


The middle part shows the '고객 검색/삭제' form with a table of customer data and input fields for search and registration. The bottom part shows the main application window with buttons for '신규 고객 등록', '고객 검색 및 수정', '전체 고객 조회', and '프로그램 종료'.

STEP 01


고객등록과 고객 검색/삭제 사용자 정의 폼 작성하기

사용자 정의 폼을 두 개 삽입하여 신규 고객을 등록하는 폼과 [고객DB] 시트에 저장된 고객 정보를 검색하고 수정/삭제하는 폼으로 설정한 후 탭 순서를 지정해보겠습니다.

01 사용자 정의 폼 추가하기 실습 파일을 엽니다. [Visual Basic 편집기]에서 ① [삽입]-[사용자 정의 폼]을 두 번 클릭합니다. 사용자 정의 폼이 두 개 추가됩니다. ② [UserForm1]을 클릭한 후 폼의 오른쪽 하단에 있는 크기 조절점을 드래그하여 크기를 변경합니다. ③ [속성] 창에서 [이름]에 frm등록, [Caption]에 고객등록을 입력합니다.



02 고객등록 폼에 컨트롤 추가하고 속성 설정하기 [고객등록] 폼에 다음과 같이 컨트롤을 추가하고 각 컨트롤의 속성을 변경합니다.



실력향상

프레임 안에 있는 컨트롤은 프레임 안에서 먼저 추가한 후에 추가해야 합니다.

1. '기본사항' 프레임

번호	컨트롤 종류	속성
1	프레임	• Caption : 기본사항
2	레이블 3개	• Caption : 주택유형, 계약일자, 공사일자
3	콤보 상자	• 이름 : cbo유형
4	텍스트 상자 2개	• 이름 : txt계약일자 • 이름 : txt공사일자


2. '고객정보' 프레임

번호	컨트롤 종류	속성
1	프레임	• Caption : 고객정보
2	레이블 7개	• Caption : 성명, 휴대폰번호, (* 하이픈 포함 입력), 주소, email, 특기사항, 고객사진 • 괄호가 있는 레이블은 Font 변경 : 크기(8)
3	텍스트 상자 6개	• 이름 : txt성명, txt번호, txt주소, txt메일, txt특기, txt사진파일명 • Font(txt사진파일명) : 글꼴 크기(8) • Locked(txt사진파일명) : True
4	명령 단추	• 이름 : cmd사진찾기 • Caption : 사진 찾기
5	이미지	• 이름 : img사진 • PictureSizeMode : 3-fromPictureModeZoom

3. '리모델링 항목' 프레임

번호	컨트롤 종류	속성
1	프레임	• Caption : 리모델링 항목
2	확인란 7개	• 이름 : chk거실, chk주방, chk안방, chk작은방, chk욕실, chk현관, chk발코니 • Caption : 거실, 주방, 안방, 작은방, 욕실, 현관, 발코니

4. '금액' 프레임




번호	컨트롤 종류	속성
1	프레임	• Caption : 금액
2	레이블 2개	• Caption : 공사비, 결제방법
3	텍스트 상자	• 이름 : txt공사비
4	옵션 단추 2개	• 이름 : opt현금, opt카드 • Caption : 현금, 카드

5. 명령 단추

신규 등록	닫 기	입력 초기화	고객DB 조회	고객 검색
-------	-----	--------	---------	-------


컨트롤 종류	속성
명령 단추 5개	<ul style="list-style-type: none"> • 이름 : cmd등록, cmd닫기, cmd초기화, cmd고객명단, cmd검색 • Caption : 신규등록, 닫기, 입력초기화, 고객DB조회, 고객검색

03 탭 순서 지정하기 ① 폼의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 ② [탭 순서]를 클릭합니다. ③ [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [Frame1] → [Frame2] → [Frame3] → [Frame4] → [cmd등록] → [cmd닫기] → [cmd초기화] → [cmd고객명단] → [cmd검색] 순서로 설정합니다. ④ [확인]을 클릭합니다.




실력향상 프레임 안에 있는 컨트롤들은 각 프레임마다 따로 탭 순서를 지정해야 합니다.


04 ① ‘기본사항’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 ② [탭 순서]를 클릭합니다. ③ [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [cbo유형] → [txt계약일자] → [txt공사일자] 순서로 설정합니다. 레이블은 탭으로 이동할 필요가 없으므로 [txt공사일자] 뒤로 이동합니다. ④ [확인]을 클릭합니다.




05 ① ‘고객정보’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 ② [탭 순서]를 클릭합니다. ③ [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [txt성명] → [txt번호] → [txt주소] → [txt메일] → [txt특기] → [cmd사진찾기] → [txt사진파일명] 순서로 설정합니다. 나머지 컨트롤은 탭으로 이동할 필요가 없으므로 [txt사진파일명] 뒤로 이동합니다. ④ [확인]을 클릭합니다.




06 ① '리모델링 항목' 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 ② [탭 순서]를 클릭합니다. ③ [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [chk거실] → [chk주방] → [chk안방] → [chk작은방] → [chk욕실] → [chk현관] → [chk발코니] 순서로 설정합니다. ④ [확인]을 클릭합니다.




07 ① '금액' 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 ② [탭 순서]를 클릭합니다. ③ [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [txt공사비] → [opt현금] → [opt카드] 순서로 설정합니다. ④ [확인]을 클릭합니다.



08 검색/삭제 폼 속성 변경하기 ① [UserForm2]를 클릭한 후 폼의 오른쪽 하단에 있는 크기 조절점을 드래그하여 크기를 변경합니다. ② [속성] 창에서 [이름]에 **frm검색**, [Caption]에 **고객 검색/삭제**를 입력합니다.



09 고객 검색/삭제 폼에 컨트롤 추가하고 속성 설정하기 [고객 검색/삭제] 폼에 다음과 같이 컨트롤을 추가하고 각 컨트롤의 속성을 변경합니다.



1. 검색어 입력 부분

번호	컨트롤 종류	속성
1	레이블 4개	<ul style="list-style-type: none"> • Caption : 검색 성명, 휴대폰번호, (1글자 이상 입력), (뒷자리 4글자 입력) • 괄호가 있는 레이블은 Font 변경 : 크기(8)
2	텍스트 상자 2개	<ul style="list-style-type: none"> • 이름 : txt검색성명, txt검색번호
3	명령 단추	<ul style="list-style-type: none"> • 이름 : cmd검색 • Caption : 검색하기


2. 검색 결과 표시 부분

번호	컨트롤 종류	속성
1	레이블	<ul style="list-style-type: none"> • Caption : 성 명 주택 유형 휴대폰 번호
2	목록 상자	<ul style="list-style-type: none"> • 이름 : lst검색 • ColumnCount : 4 • ColumnWidths : 40pt; 60pt; 70pt; 0pt • ListStyle : 1-fmListStyleOption
3	명령 단추 3개	<ul style="list-style-type: none"> • 이름 : cmd등록, cmd고객DB, cmd닫기 • Caption : 신규 등록, 고객 DB, 닫 기


3. '고객상세정보' 프레임

번호	컨트롤 종류	속성
1	프레임	<ul style="list-style-type: none"> • Caption : 고객상세정보
2	레이블 12개	<ul style="list-style-type: none"> • Caption : 성 명, 주택유형, 계약일자, 공사 일자, 휴대폰번호, 공사비, 결제방법, 주소, email, 특기 사항, 리모델링 항목, 사진
3	텍스트 상자 10개	<ul style="list-style-type: none"> • 이름 : txt조회성명, txt조회유형, txt조회계약 일자, txt조회공사일자, txt조회번호, txt조회 공사비, txt조회결제방법, txt조회주소, txt조회메일, txt조회특기
4	목록 상자	<ul style="list-style-type: none"> • 이름 : lst조회항목
5	이미지	<ul style="list-style-type: none"> • 이름 : img조회사진
6	명령 단추 2개	<ul style="list-style-type: none"> • 이름 : cmd수정, cmd삭제 • Caption : 수정 사항 등록, 고객 삭제

10 탭 순서 지정하기 ① ‘고객상세정보’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 ② [탭 순서]를 클릭합니다. ③ [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [txt검색성명] → [txt검색번호] → [lst검색] → [cmd검색] → [Frame1] → [cmd등록] → [cmd고객DB] → [cmd닫기] 순서로 설정합니다. ④ [확인]을 클릭합니다.



11 ① ‘고객상세정보’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 ② [탭 순서]를 클릭합니다. ③ [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [txt조회성명] → [txt조회유형] → [txt조회계약일자] → [txt조회공사일자] → [txt조회번호] → [txt조회공사비] → [txt조회결제방법] → [txt조회주소] → [txt조회메일] → [txt조회특기] → [txt조회항목] → [cmd수정] → [cmd삭제] 순서로 설정합니다. ④ [확인]을 클릭합니다.




STEP 02

고객등록 폼 이벤트 프로시저 작성하기


[고객등록] 폼이 실행되면서 준비해야 할 내용을 'Initialize' 이벤트 프로시저에 작성하고, [사진 찾기] 단추에는 [사진] 폴더 안에 준비된 고객 사진을 찾아오도록 이벤트 프로시저를 작성합니다. [신규 등록] 단추에는 폼에 입력된 항목들이 [고객DB] 시트에 자동 입력되도록 하는 이벤트 프로시저를, [입력 초기화] 단추에는 폼의 컨트롤들을 모두 초기화하는 이벤트 프로시저를 작성해보겠습니다.

12 Initialize 이벤트 프로시저 작성하기 ① [프로젝트] 탐색기 창에서 [폼]-[frm검색]을 더블클릭합니다. [고객등록] 폼이 나타나면 ② 폼의 빈 공간을 더블클릭합니다. [코드] 창이 표시되면 ③ [프로시저]에서 [Initialize]를 선택합니다.



실력향상 Initialize 이벤트는 폼이 실행되어 화면에 보여지기 바로 전에 한 번 발생하는 이벤트입니다.

13 'UserForm_Initialize' 이벤트 프로시저가 생성됩니다. ① 'UserForm_Click' 이벤트 프로시저는 범위로 지정하고 ② [Delete]를 눌러 삭제합니다. ③ 'UserForm_Initialize' 이벤트 프로시저에 주택유형 항목을 추가하고 계약 날짜로 오늘 날짜를 표시하는 코드를 작성합니다.




```

1 Private Sub UserForm_Initialize()
2     With Me.cbo유형
3         .AddItem "아파트"
4         .AddItem "단독주택"
5         .AddItem "연립주택"
6         .AddItem "다가구주택"
7     End With
9     Me.txt계약일자.Value = Date
End Sub


```

- ① 현재 폼이 실행되기 바로 전에 발생하는 'Initialize' 이벤트 프로시저를 시작합니다.
- ② Me.cbo유형을 With 문으로 묶어서 ③ ~ ⑥의 마침표(.) 앞에 생략되었음을 표시합니다.
- ③ ~ ⑥ 'cbo유형' 콤보 상자에 항목을 하나씩 추가합니다. AddItem 메서드는 콤보 상자에 행 목록을 추가합니다.
- ⑦ With 문을 종료합니다.
- ⑧ 'txt계약일자' 텍스트 상자에 오늘 날짜를 입력합니다. VBA에서 Date 함수는 오늘 날짜를 표시합니다.


14 결과 확인하기 ① [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행]을 클릭합니다. 폼이 실행되면 ② 'cbo유형' 콤보 상자의 목록 단추를 클릭합니다. 추가한 항목이 표시됩니다. 'txt계약일자' 텍스트 상자에는 오늘 날짜가 표시됩니다.



15 사진 찾기 단추의 Click 이벤트 프로시저 작성하기 ‘고객정보’ 프레임에 있는 ‘cmd사진찾기’ 명령 단추를 더블클릭합니다. ‘cmd사진찾기_Click’ 이벤트 프로시저가 추가됩니다.




16 ‘cmd사진찾기_Click’ 이벤트 프로시저에 사진 파일을 선택할 수 있는 열기 대화상자를 표시한 후 사용자가 선택하는 사진을 이미지 컨트롤에 표시하고 사진 파일 이름은 텍스트 상자에 표시하도록 하는 코드를 입력합니다.




- ① Private Sub cmd사진찾기_Click()
 - ② Dim 사진파일 As Variant
 - ③ 사진파일 = Application.GetOpenFilename(Filefilter:="그림파일,*.gif;*.jpg;*.bmp", Title:="고객 사진 선택")
 - ④ If 사진파일 = False Then Exit Sub
 - ⑤ Me.img사진.Picture = LoadPicture(사진파일)
 - ⑥ Me.txt사진파일명 = Dir(사진파일)
- End Sub

- ① 'cmd사진 찾기' 명령 단추를 클릭했을 때 발생할 이벤트 프로시저를 시작합니다.
- ② '사진파일' 변수를 가변형으로 선언합니다. 이 변수에는 열기 대화상자에서 선택하는 사진 파일의 정보를 저장합니다. 열기 대화상자에서 사진을 선택하면 선택한 사진의 경로를 포함한 파일명을 저장하고, 만약 열기 대화상자에서 [취소]를 클릭하면 변수에 False를 저장하므로 변수를 선언할 때 자료 형식을 가변형으로 선언합니다.
- ③ 한 줄로 입력합니다. 사진 파일을 선택할 수 있는 열기 대화상자를 표시합니다. 열기 대화상자의 이름은 '고객 사진 선택'으로 표시하고, gif, jpg, bmp 파일 형식을 선택할 수 있도록 설정됩니다. 사용자가 선택하는 사진의 파일 정보를 '사진파일' 변수에 저장합니다.
- ④ '사진파일' 변수가 False일 경우 프로시저를 종료하는 조건문입니다. 한 줄로 입력하는 If 문으로, End If가 없습니다. 만약 열기 대화상자에서 [취소]를 클릭하면 '사진파일' 변수에 False가 저장되어 ⑤에서 오류가 발생하므로 [취소]를 클릭하면 프로시저를 종료합니다.
- ⑤ 선택한 사진 파일을 이미지 컨트롤에 표시합니다.
- ⑥ 선택한 사진 파일의 이름을 'txt사진파일명' 텍스트 상자에 대입합니다. 이 텍스트 상자에 입력된 파일 이름은 [고객DB] 시트에 저장할 때 사용합니다.


17 결과 확인하기 ① [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행]을 클릭합니다. 폼이 실행되면 ② 'cmd사진찾기' 명령 단추를 클릭합니다.




18 ① [사진] 폴더 안에 준비된 사진 중 하나를 클릭한 후 ② [열기]를 클릭합니다. 이미지 컨트롤에 사진이 표시되고, 'txt사진파일명' 텍스트 상자에 사진 파일 이름이 표시됩니다.



19 신규 등록 단추의 Click 이벤트 프로시저 작성하기 'cmd등록' 명령 단추를 더블클릭합니다. 'cmd등록_Click' 이벤트 프로시저가 추가됩니다.



20 [고객등록] 대화상자에서 [신규 등록]을 클릭하면 입력한 값들이 [고객DB] 시트에 입력되도록 하는 코드를 입력합니다.



- ① Private Sub cmd등록_Click()
- ② Dim 행번호 As Integer
- ③ 행번호 = Sheets("고객DB").Range("A1").CurrentRegion.Rows.Count + 1
- ④ With Sheets("고객DB")
- ⑤ .Cells(행번호, 1) = Format(Me.txt계약일자, "mmdd") & "-" & Right(Me.txt번호, 4)
- ⑥ .Cells(행번호, 2) = Me.cbo유형
- ⑦ .Cells(행번호, 3) = Me.txt계약일자
- ⑧ .Cells(행번호, 4) = Me.txt공사일자
- ⑨ .Cells(행번호, 5) = Me.txt성명
- ⑩ .Cells(행번호, 6) = Me.txt번호
- ⑪ .Cells(행번호, 7) = Me.txt주소
- ⑫ .Cells(행번호, 8) = Me.txt메일
- ⑬ .Cells(행번호, 9) = Me.txt특기
- ⑭ .Cells(행번호, 10) = Me.txt사진파일명
- ⑮ If Me.chk거실 Then .Cells(행번호, 11).Value = "●"
- ⑯ If Me.chk주방 Then .Cells(행번호, 12).Value = "●"
- ⑰ If Me.chk안방 Then .Cells(행번호, 13).Value = "●"
- ⑱ If Me.chk작은방 Then .Cells(행번호, 14).Value = "●"


```

19      If Me.chk욕실 Then .Cells(행번호, 15).Value = "●"
20      If Me.chk현관 Then .Cells(행번호, 16).Value = "●"
21      If Me.chk발코니 Then .Cells(행번호, 17).Value = "●"
22      .Cells(행번호, 18) = Val(Me.txt공사비)
23      If Me.opt현금 Then
24          .Cells(행번호, 19) = "현금"
25      ElseIf Me.opt카드 Then
26          .Cells(행번호, 19) = "카드"
27      End If
28  End With
29      MsgBox Me.txt성명 & "님 리모델링 정보가 등록되었습니다"
End Sub


```

- ① 'cmd등록' 명령 단추를 클릭했을 때 실행할 이벤트 프로시저를 시작합니다.
- ② [고객DB] 시트에 입력할 셀 위치의 행 번호를 저장할 '행번호' 변수를 선언합니다.
- ③ [고객DB] 시트의 [A1] 셀을 기준으로 데이터가 입력된 행 개수를 계산하고 그 결과에 1을 더한 값을 '행번호' 변수에 대입합니다.
- ④ **Sheets("고객DB")**를 With 문으로 묶어 ⑤ ~ ⑳의 마침표(.) 앞에 생략되었음을 표시합니다.
- ⑤ A열에 고객코드를 '계약일자-월일-휴대폰번호 뒷자리 4글자' 형식으로 만들어 입력합니다.
- ⑥ ~ ⑭ [B:] 열에 주택유형, 계약일자, 공사일자, 성명, 휴대폰번호, 주소, email, 특기사항, 고객 사진 파일 이름을 순서대로 입력합니다.
- ⑮ ~ ㉑ [K:Q] 열에 리모델링 항목의 확인란이 선택되어 있으면 각 열에 ●를 입력합니다. 한 줄로 입력하는 If 조건문으로 End If가 없습니다.
- ㉒ R열에 공사비를 숫자 형식으로 변환하여 입력합니다.
- ㉓ ~ ㉗ S열에 결제 방법을 입력하는 조건문입니다. 현금이 선택되어 있으면 **현금**, 카드가 선택되어 있으면 **카드**를 입력합니다.
- ㉘ With 문을 종료합니다.
- ㉙ 등록이 완료되었다는 메시지 대화상자를 나타냅니다.

21 입력 초기화 단추의 Click 이벤트 프로시저 작성하기 [개체]에서 [cmd초기화]를 클릭합니다.
 'cmd초기화_Click' 이벤트 프로시저가 추가됩니다.



22 [고객등록] 대화상자에서 [입력 초기화]를 클릭하면 입력하거나 선택한 값이 모두 초기화되는 코드를 입력합니다.



- ① Private Sub cmd초기화_Click()
- ② Me.cbo유형.ListIndex = -1
- ③ Me.txt계약일자 = Date
- ④ Me.txt공사일자 = ""
- ⑤ Me.txt성명 = ""
- ⑥ Me.txt번호 = ""
- ⑦ Me.txt주소 = ""
- ⑧ Me.txt메일 = ""


```

9      Me.txt특기 = ""
10     Me.img사진.Picture = Nothing
11     Me.txt사진파일명 = ""
12     Me.chk거실 = False
13     Me.chk주방 = False
14     Me.chk안방 = False
15     Me.chk작은방 = False
16     Me.chk욕실 = False
17     Me.chk현관 = False
18     Me.chk발코니 = False
19     Me.txt공사비 = ""
20     Me.opt현금 = False
21     Me.opt카드 = False
End Sub

```

- ① ‘cmd초기화’ 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저를 시작합니다.
- ② ‘cbo유형’ 콤보 상자에 선택된 내용을 지웁니다. 콤보 상자의 ListIndex 값을 -1로 설정하면 선택된 항목이 없는 것을 뜻합니다.
- ③ ‘txt계약일자’ 텍스트 상자에 오늘 날짜를 입력합니다. 계약일자는 폼이 실행되기 전 Initialize 이벤트에 의해 오늘 날짜가 표시되므로 초기화 작업으로 오늘 날짜를 입력합니다.
- ④ ~ ⑨ ‘txt공사일자’, ‘txt성명’, ‘txt번호’, ‘txt주소’, ‘txt이메일’, ‘txt특기’ 텍스트 상자에 빈 값을 대입합니다.
- ⑩ 사진을 지웁니다.
- ⑪ ‘txt사진파일명’ 텍스트 상자에 빈 값을 대입합니다.
- ⑫ ~ ⑮ ‘리모델링 항목’을 모두 선택 해제합니다.
- ⑯ ‘txt공사비’ 텍스트 상자에 빈 값을 대입합니다.
- ⑰ ~ ⑳ ‘결제방법’ 항목인 ‘opt현금’과 ‘opt카드’의 선택을 모두 해제합니다.

23 달기, 고객DB조회, 고객 검색 단추의 Click 이벤트 프로시저 작성하기 같은 방법으로 [개체]에서 [cmd달기], [cmd고객명단], [cmd검색]을 각각 클릭하여 'Click' 이벤트 프로시저를 작성합니다.



```
1 Private Sub cmd검색_Click()
2     Unload Me
3     frm검색.Show vbModeless
End Sub
```

- ❶ 'cmd검색' 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저입니다.
- ❷ 현재 폼을 닫습니다.
- ❸ 검색 폼을 모달리스 형식으로 실행합니다.

```
1 Private Sub cmd고객명단_Click()
2     Sheets("고객DB").Select
3     Unload Me
End Sub
```

- ❶ 'cmd고객명단' 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저입니다.
- ❷ [고객DB] 시트를 선택합니다.
- ❸ 현재 폼을 닫습니다.

```


1 Private Sub cmd닫기_Click()
2     Unload Me
End Sub

```

① 'cmd닫기' 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저입니다.

② 현재 폼을 닫습니다.

24 필수 항목 입력 확인하기 고객 정보를 등록할 때 주택유형, 성명, 휴대폰번호, 리모델링 항목은 필수 항목으로 꼭 입력하도록 'cmd등록' 명령 단추의 'Click' 이벤트 프로시저를 수정합니다.



```

Private Sub cmd등록_Click()
Dim 행번호 As Integer
행번호 = Sheets("고객DB").Range("A1").CurrentRegion.Rows.Count + 1
1 If Me.cbo유형.ListIndex = -1 Then
2     MsgBox "주택유형이 선택되지 않았습니다"
3     cbo유형.SetFocus
4     SendKeys "{F4}"
5     Exit Sub
6 End If
7 If Me.txt성명 = "" Then
8     MsgBox "성명이 입력되지 않았습니다"

```

```

9      Me.txt성명.SetFocus
10     Exit Sub
11 End If
12 If Me.txt번호 = "" Then
13     MsgBox "휴대폰 번호가 입력되지 않았습니다"
14     Me.txt번호.SetFocus
15     Exit Sub
16 End If
17 If Me.chk거실 = False And Me.chk주방 = False And Me.chk안방 = False And Me.chk
    작은방 = False And Me.chk욕실 = False And Me.chk현관 = False And Me.chk
    발코니 = False Then
18     MsgBox " 리모델링 항목이 선택되지 않았습니다"
19     Exit Sub
20 End If
    With Sheets("고객DB")
        .Cells(행번호, 1) = Format(Me.txt계약일자, "mmdd") & "-" & Right(Me.txt번호, 4)
        .Cells(행번호, 2) = Me.cbo유형
        .Cells(행번호, 3) = Me.txt계약일자
        .Cells(행번호, 4) = Me.txt공사일자
        .Cells(행번호, 5) = Me.txt성명
        .Cells(행번호, 6) = Me.txt번호
        .Cells(행번호, 7) = Me.txt주소
        .Cells(행번호, 8) = Me.txt메일
        .Cells(행번호, 9) = Me.txt특기
        .Cells(행번호, 10) = Me.txt사진파일명
        If Me.chk거실 Then .Cells(행번호, 11).Value = "●"
        If Me.chk주방 Then .Cells(행번호, 12).Value = "●"
        If Me.chk안방 Then .Cells(행번호, 13).Value = "●"
        If Me.chk작은방 Then .Cells(행번호, 14).Value = "●"
        If Me.chk욕실 Then .Cells(행번호, 15).Value = "●"
        If Me.chk현관 Then .Cells(행번호, 16).Value = "●"
        If Me.chk발코니 Then .Cells(행번호, 17).Value = "●"
        .Cells(행번호, 18) = Val(Me.txt공사비)
        If Me.opt현금 Then
            .Cells(행번호, 19) = "현금"
        ElseIf Me.opt카드 Then
            .Cells(행번호, 19) = "카드"
        End If
    End With
End With

```

MsgBox Me.txt성명 & "님 리모델링 정보가 등록되었습니다"

21 Call cmd초기화_Click

End Sub

① ~ ⑥ 주택유형을 선택하지 않으면 메시지 대화상자를 나타내고, 'cho유형' 콤보 상자의 목록을 펼쳐서 표시한 후 프로시저를 종료합니다. **SendKeys "{F4}"**는 키보드의 **[F4]**를 누르는 것과 같습니다.


⑦ ~ ⑪ 성명을 입력하지 않으면 메시지 대화상자를 나타내고, 'txt성명' 텍스트 상자로 커서를 이동시킨 후 프로시저를 종료합니다.

⑫ ~ ⑯ 휴대폰번호를 입력하지 않으면 메시지 대화상자를 나타내고, 'txt번호' 텍스트 상자로 커서를 이동시킨 후 프로시저를 종료합니다.


⑰ ~ ⑳ '리모델링 항목' 프레임에 있는 확인란은 최소 한 개 이상 선택되어야 하는데 모두 선택되지 않으면 메시지 대화상자를 나타내고 프로시저를 종료합니다. ⑰은 한 줄로 입력합니다.

㉑ 'cmd초기화_Click' 이벤트 프로시저를 호출하여 실행합니다. [고객DB] 시트에 내용을 모두 입력한 후 컨트롤들을 초기화합니다.

25 결과 확인하기 ① [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행]을 클릭합니다. 폼이 실행되면 ② 각 컨트롤에 내용을 입력하거나 선택한 후 [신규 등록]을 클릭합니다. 등록되었다는 메시지가 나타나면 ③ [확인]을 클릭합니다.



26 [고객DB] 시트에 입력된 내용을 확인합니다.




27 다시 폼을 실행한 후 아무것도 입력하지 않은 상태에서 ❶ [신규 등록]을 클릭합니다. 주택유형이 선택되지 않았다는 메시지가 나타나고 ❷ [확인]을 클릭하면 [주택유형] 목록이 펼쳐져 표시됩니다.

STEP 03


고객 검색/삭제 폼의 이벤트 프로시저 작성하기

[고객 검색/삭제] 폼에서 검색어를 입력한 후 [검색하기]를 클릭하면 [고객DB] 시트에서 고객 정보를 검색하여 목록 상자에 표시하고, 목록 상자에서 한 명의 고객을 클릭하면 고객상세정보가 표시되도록 이벤트 프로시저를 작성해보겠습니다. 또한 고객 정보를 수정하거나 삭제하면 [고객DB] 시트에 반영되도록 이벤트 프로시저를 작성합니다.

28 검색하기 단추의 Click 이벤트 프로시저 작성하기 ① [프로젝트] 탐색기 창에서 [폼]-[frm검색]을 더블클릭합니다. 폼이 표시되면 ② 'cmd검색' 명령 단추를 더블클릭합니다. [코드] 창이 표시되고 'cmd검색_Click' 이벤트 프로시저가 추가됩니다.



29 성명과 휴대폰번호로 고객 정보 검색하기 'txt검색성명'과 'txt검색번호'의 텍스트 상자에 한 글자 이상 입력한 후 [검색하기]를 클릭하면 [고객DB] 시트에서 일치하는 고객 정보를 찾아 목록 상자에 표시하는 문장을 입력합니다.




```


1 Private Sub cmd검색_Click()
2 Dim 셀위치 As Range
3 Dim 행번호 As Integer
4 Set 셀위치 = Sheets("고객DB").Range("E2")
5 행번호 = 0
6 Me.lst검색.Clear
7 Do While 셀위치.Value <> ""
8     If 셀위치.Value Like "*" & Me.txt검색성명 & "*" And 셀위치.Offset(0, 1).Value
        Like "*" & Me.txt검색번호 & "*" Then
9         Me.lst검색.AddItem 셀위치.Value
10        Me.lst검색.Column(1, 행번호) = 셀위치.Offset(0, -3)
11        Me.lst검색.Column(2, 행번호) = 셀위치.Offset(0, 1)
12        Me.lst검색.Column(3, 행번호) = 셀위치.Offset(0, -4)
13        행번호 = 행번호 + 1
14    End If
15    Set 셀위치 = 셀위치.Offset(1, 0)
16 Loop
End Sub

```


- ① ‘cmd검색’ 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저를 시작합니다.
- ② [고객DB] 시트에서 검색할 성명의 셀 위치를 저장할 ‘셀위치’ 변수를 선언합니다.
- ③ 목록 상자에 추가되는 행 목록 수를 저장할 ‘행번호’ 변수를 선언합니다.
- ④ ‘셀위치’ 변수에 [고객DB] 시트의 [E2] 셀을 대입합니다. [E2] 셀에는 등록된 고객의 첫 번째 성명이 있습니다.
- ⑤ ‘행번호’ 변수에 0을 대입합니다. 목록 상자의 ListIndex 시작값이 0이기 때문에 0을 대입합니다.
- ⑥ ‘lst검색’ 목록 상자를 초기화하는 문장으로, 목록 상자에 표시된 항목이 있다면 모두 제거합니다.
- ⑦ ‘셀위치’ 변수에 저장된 셀에 값이 있는 동안 ⑧ ~ ⑮를 반복 실행합니다.
- ⑧ 조건문을 시작하여 조건식이 참이면 ⑨ ~ ⑬을 실행합니다. ‘txt검색성명’ 텍스트 상자의 값과 [고객DB] 시트의 성명을 비교하고, ‘txt검색’ 텍스트 상자의 값과 [고객DB] 시트의 휴대폰번호를 비교합니다. 별표(*)를 이용하여 포함된 문자를 비교할 때는 Like 연산자를 사용합니다. 한 줄로 입력합니다.
- ⑨ ‘lst검색’ 목록 상자에 성명을 추가합니다. 첫 번째 열을 추가할 때는 AddItem 메서드를 사용합니다.
- ⑩ ~ ⑫ 추가한 성명과 같은 ListIndex에 주택유형, 휴대폰번호, 고객코드를 열 단위로 추가합니다. 두 번째 열에 항목을 추가할 때는 Column 속성을 사용합니다. 셀위치.Offset(0, -3)는 성명이 입력된 열을 기준으로 왼쪽으로 세 칸 이동한 셀의 값을 뜻하므로 주택유형이 됩니다.
- ⑬ 목록 상자의 목록이 한 개 추가되었으므로 ‘행번호’ 변수를 1 증가시킵니다.

- 14 If 조건문을 종료합니다.
- 15 다음 성명을 비교하기 위해 아래쪽으로 한 칸 이동한 셀을 '셀위치' 변수에 대입합니다.
- 16 반복문을 종료합니다.

30 결과 확인하기 ① [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행]을 클릭합니다. 폼이 실행되면 ② [검색 성명]에 **김**을 입력한 후 ③ [검색하기]를 클릭합니다. 조건에 맞는 고객 정보가 검색되어 목록 상자에 표시됩니다. ④ [닫기]를 클릭해 폼을 닫습니다.




31 선택한 고객의 셀 위치를 찾는 프로시저 작성하기 ① [삽입]-[모듈]을 클릭합니다. ② 모듈이 추가되면 [속성] 창에서 [이름]을 **검색프로그램**으로 변경합니다.



실력향상 목록 상자에 표시된 고객 중 한 명을 선택했을 때 그 정보가 오른쪽 [고객상세정보] 프레임에 표시되도록 하려면 선택한 고객의 정보가 [고객DB] 시트의 어디에 입력되어 있는지 알아야 합니다. 이 기능은 고객 정보를 수정하거나 삭제할 때도 필요한 기능으로, 공용 모듈에 작성합니다. 공용 모듈에 작성한 프로시저는 현재 통합 문서에서 필요할 때마다 Call 문으로 호출하여 사용할 수 있습니다.

32 현재 통합 문서에서 모두 사용할 수 있는 프로젝트 수준의 변수를 선언하고 목록 상자에서 선택한 고객 정보가 [고객DB] 시트의 어디에 입력되어 있는지 셀 위치를 찾아내는 ‘고객정보검색’ 프로시저를 작성합니다.




- ① Public 고객코드 As String
 - ② Public 고객범위 As Range
 - ③ Public 셀위치 As Range
 - ④ Sub 고객정보검색()
 - ⑤ On Error GoTo 에러처리
 - ⑥ 고객코드 = frm검색.lst검색.Column(3, frm검색.lst검색.ListIndex)
 - ⑦ Set 고객범위 = Sheets("고객DB").Range("A2", Sheets("고객DB").Range("A2").End(xlDown))
 - ⑧ Set 셀위치 = 고객범위.Find(고객코드, Lookat:=xlWhole)
 - ⑨ Exit Sub
 - ⑩ 에러처리:
 - ⑪ MsgBox "목록에서 고객을 먼저 선택하세요"
- End Sub


- ① 목록 상자에서 선택한 고객의 고객코드를 저장할 '고객코드' 변수를 선언합니다. 이 변수가 사용되는 '고객정보검색' 프로시저는 [고객 검색/삭제] 폼에서 여러 번 호출하여 사용되므로 프로시저 밖에서 Public 으로 선언해야 합니다.
- ② [고객DB] 시트에 입력된 고객코드의 범위를 저장할 '고객범위' 변수를 선언합니다.
- ③ 일치하는 고객코드를 찾았을 때 셀 위치를 저장할 '셀위치' 변수를 선언합니다. 목록 상자에서 선택한 고객을 [고객DB] 시트에서 찾을 때 고객코드를 기준으로 찾아야 중복값이 없습니다.
- ④ '고객정보검색' 프로시저를 시작합니다.
- ⑤ 오류가 발생하면 '에러처리' 레이블이 있는 곳으로 이동합니다. 만약 목록 상자에서 고객이 선택되지 않은 상태에서 이 프로시저가 실행된다면 ⑥에서 오류가 발생하므로 이를 해결하기 위한 코드입니다.

- ⑥ 목록 상자에서 선택한 고객의 고객코드를 '고객코드' 변수에 대입합니다. 목록 상자의 Column 속성은 'Column(열번호, 행번호)' 형식으로 사용하는데, 열 번호는 네 번째 열이므로 **3**이고, 행 번호는 목록 상자에서 선택된 행이 바로 ListIndex이므로 ListIndex 속성값을 행 번호로 사용합니다.
- ⑦ [고객DB] 시트의 [A2] 셀에서 [Ctrl]+[↓]를 눌렀을 때 선택되는 범위를 '고객범위' 변수에 대입합니다.
- ⑧ Find 메서드를 이용하여 '고객범위' 변수에 저장된 범위에서 '고객코드' 변수에 저장된 코드와 일치하는 값을 찾아 해당 셀을 '셀위치' 변수에 대입합니다. Find 메서드의 **Lookat:=xlWhole** 인수를 설정하면 전체 셀 데이터가 일치하는 항목을 찾습니다. '셀위치' 변수를 프로시저 밖에 Public으로 선언했기 때문에 프로시저가 종료되더라도 저장된 값은 유지됩니다.
- ⑨ 프로그램이 정상적으로 실행되었을 때 ⑩ ~ ⑪이 실행되지 않도록 프로시저를 종료합니다.
- ⑩ 오류가 발생했을 때 이동할 곳입니다.
- ⑪ 목록 상자에서 고객을 먼저 선택하도록 메시지 대화상자를 나타냅니다.

33 목록 상자의 Click 이벤트 프로시저 작성하기 ① [프로젝트] 탐색기 창에서 [폼]-[frm검색]을 더블클릭합니다. ② 폼이 표시되면 'lst검색' 목록 상자를 더블클릭합니다. 'lst검색_Click' 이벤트 프로시저가 추가됩니다.



34 '1st검색' 목록 상자에서 고객 한 명을 클릭하면 상세 정보가 '고객상세정보' 프레임의 각 컨트롤에 표시되도록 하는 코드를 입력합니다.




```

1 Private Sub 1st검색_Click()
2 Call 고객정보검색
3 Me.txt조회성명 = 셀위치.Offset(0, 4)
4 Me.txt조회유형 = 셀위치.Offset(0, 1)
5 Me.txt조회계약일자 = 셀위치.Offset(0, 2)
6 Me.txt조회공사일자 = 셀위치.Offset(0, 3)
7 Me.txt조회번호 = 셀위치.Offset(0, 5)
8 Me.txt조회주소 = 셀위치.Offset(0, 6)
9 Me.txt조회메일 = 셀위치.Offset(0, 7)
10 Me.txt조회특기 = 셀위치.Offset(0, 8)
11 If 셀위치.Offset(0, 9) <> "" Then Me.img조회사진.Picture = _
    LoadPicture(ThisWorkbook.Path & "\사진\" & 셀위치.Offset(0, 9))
12 Me.lst조회항목.Clear
13 If 셀위치.Offset(0, 10) = "●" Then Me.lst조회항목.AddItem "거실"
14 If 셀위치.Offset(0, 11) = "●" Then Me.lst조회항목.AddItem "주방"
15 If 셀위치.Offset(0, 12) = "●" Then Me.lst조회항목.AddItem "안방"
16 If 셀위치.Offset(0, 13) = "●" Then Me.lst조회항목.AddItem "작은방"
17 If 셀위치.Offset(0, 14) = "●" Then Me.lst조회항목.AddItem "욕실"
18 If 셀위치.Offset(0, 15) = "●" Then Me.lst조회항목.AddItem "현관"
19 If 셀위치.Offset(0, 16) = "●" Then Me.lst조회항목.AddItem "발코니"
20 Me.txt조회공사비 = 셀위치.Offset(0, 17)
21 Me.txt조회결제방법 = 셀위치.Offset(0, 18)
End Sub


```

- ① '1st검색' 목록 상자를 클릭했을 때 발생할 이벤트 프로시저를 시작합니다.
- ② '고객정보검색' 프로시저를 호출합니다. 이 프로시저는 [검색프로그램] 모듈에 작성되어 있으며 이 프로시저가 실행되고 나면 '셀위치' 변수에는 목록 상자에서 선택한 고객의 고객코드가 [고객DB] 시트의 어느 셀에 있는지 셀 정보를 가지고 복귀합니다.
- ③ ~ ⑩ 목록 상자에서 선택한 고객의 정보를 '고객상세정보' 프레임에 있는 텍스트 상자에 표시합니다. '셀위치' 변수에는 [고객DB] 시트의 A열 중 한 셀의 정보가 저장되어 있으므로 성명, 주택유형 등의 정보는 A열을 기준으로 Offset 속성으로 열을 이동하며 해당 항목의 값을 가져옵니다.
- ⑪ 이미지 컨트롤에 사진 파일을 표시합니다. 사진 파일은 현재 통합 문서가 있는 폴더 안의 [사진] 폴더에 반드시 저장되어 있어야 합니다. 한문장인데 언더바(_)를 사용하여 두 줄로 나누어 작성했습니다.
- ⑫ '리모델링 항목'의 '1st조회항목' 목록 상자에 새로운 정보를 표시하기 위해 기존에 있는 목록을 모두 삭제합니다.
- ⑬ ~ ⑰ 각 리모델링 항목 열에 ●가 표시되어 있으면 해당 항목을 목록 상자에 추가합니다.
- ⑱ 공사비를 'txt조회공사비' 텍스트 상자에 표시합니다.
- ㉑ 결제방법을 'txt조회결제방법' 텍스트 상자에 표시합니다.

35 수정 사항 등록 단추의 Click 이벤트 프로시저 작성하기 [개체]에서 [cmd수정]을 클릭합니다. 'cmd수정_Click' 이벤트 프로시저가 추가됩니다.



36 [수정 사항 등록]을 클릭하면 ‘고객상세정보’ 프레임 안에 있는 각 컨트롤의 값이 [고객DB] 시트에 업데이트되도록 하는 코드를 입력합니다.




- ① Private Sub cmd수정_Click()
- ② Call 고객정보검색
- ③ 셀위치.Offset(0, 4) = Me.txt조회성명
- ④ 셀위치.Offset(0, 1) = Me.txt조회유형
- ⑤ 셀위치.Offset(0, 2) = Me.txt조회계약일자
- ⑥ 셀위치.Offset(0, 3) = Me.txt조회공사일자
- ⑦ 셀위치.Offset(0, 5) = Me.txt조회번호
- ⑧ 셀위치.Offset(0, 6) = Me.txt조회주소
- ⑨ 셀위치.Offset(0, 7) = Me.txt조회메일
- ⑩ 셀위치.Offset(0, 8) = Me.txt조회특기
- ⑪ 셀위치.Offset(0, 17) = Val(Me.txt조회공사비)
- ⑫ 셀위치.Offset(0, 18) = Me.txt조회결제방법
- ⑬ Call cmd검색_Click
- ⑭ MsgBox Me.txt조회성명 & "님의 정보가 수정되었습니다"
- End Sub

- ① ‘cmd수정’ 명령 단추를 클릭했을 때 발생하는 이벤트 프로시저를 시작합니다.
- ② ‘고객정보검색’ 프로시저를 호출합니다. 이 프로시저는 [검색프로그램] 모듈에 작성되어 있으며 이 프로시저가 실행되고 나면 ‘셀위치’ 변수에는 목록 상자에서 선택한 고객의 고객코드가 [고객DB] 시트의 어느 셀에 있는지 셀 정보를 가지고 복귀합니다.
- ③ ~ ⑫ 각 텍스트 상자의 현재값을 [고객DB] 시트의 해당 셀에 다시 입력합니다.


13 변경된 데이터로 다시 검색하기 위하여 'cmd검색_Click' 이벤트 프로시저를 실행합니다.

14 정보가 수정되었다는 메시지 대화상자를 나타냅니다.

37 고객 삭제 단추의 Click 이벤트 프로시저 작성하기 [개체]에서 [cmd삭제]를 클릭합니다. 'cmd삭제_Click' 이벤트 프로시저가 추가됩니다.



38 [고객 삭제]를 클릭하면 선택된 고객 정보가 [고객DB] 시트에서 삭제되도록 하는 코드를 입력 합니다.



- 1 Private Sub cmd삭제_Click()
- 2 If Me.txt조회성명 = "" Then
- 3 MsgBox "삭제할 고객이 선택되지 않았습니다"


```

④      Exit Sub
⑤      End If
⑥      If MsgBox(Me.txt조회성명 & "님의 정보를 삭제하시겠습니까?", vbYesNo) = vbNo Then
          Exit Sub
⑦      Call 고객정보검색
⑧      셀워치.EntireRow.Delete (xlUp)
⑨      Me.txt조회성명 = ""
⑩      Me.txt조회유형 = ""
⑪      Me.txt조회계약일자 = ""
⑫      Me.txt조회공사일자 = ""
⑬      Me.txt조회번호 = ""
⑭      Me.txt조회공사비 = ""
⑮      Me.txt조회결제방법 = ""
⑯      Me.txt조회주소 = ""
⑰      Me.txt조회메일 = ""
⑱      Me.txt조회특기 = ""
⑲      Me.lst조회항목.Clear
⑳      Me.img조회사진.Picture = Nothing
㉑      Call cmd검색_Click
End Sub

```

- ① ‘cmd삭제’ 명령 단추를 클릭했을 때 발생하는 이벤트 프로시저를 시작합니다.
- ② 삭제할 고객이 선택되었는지 확인하는 조건문입니다. 만약 목록 상자에서 삭제할 고객이 선택되지 않은 상태라면 ‘고객상세정보’ 프레임에 있는 ‘txt조회성명’ 텍스트 상자가 비어 있습니다. 조건식이 참이면 ③ ~ ④를 실행합니다.
- ③ ~ ④ 조건식이 참일 때 실행할 문장으로, 삭제할 고객이 선택되지 않았다는 메시지 대화상자를 나타내고 프로시저를 종료합니다.
- ⑤ If 조건문을 종료합니다.
- ⑥ 한 줄로 입력합니다. 삭제 진행여부를 확인하는 메시지 대화상자를 나타내고, [아니오]를 클릭하면 프로시저가 종료됩니다. 한 줄에 입력한 If 조건문으로, End If 문이 없습니다.
- ⑦ ‘고객정보검색’ 프로시저를 호출합니다. 이 프로시저는 [검색프로그램] 모듈에 작성되어 있으며 이 프로시저가 실행되고 나면 ‘셀워치’ 변수에는 목록 상자에서 선택한 고객의 고객코드가 [고객DB] 시트의 어느 셀에 있는지 셀 정보를 가지고 복귀합니다.
- ⑧ ‘셀워치’ 변수에 저장된 셀의 행 전체를 삭제합니다.
- ⑨ ~ ㉑ ‘상세고객정보’ 프레임의 컨트롤들을 모두 초기화합니다.
- ㉑ 삭제된 결과를 확인하기 위하여 ‘cmd검색_Click’ 이벤트 프로시저를 실행합니다.

39 신규 등록, 고객 DB, 닫기 단추의 Click 이벤트 프로시저 작성하기 같은 방법으로 [개체]에서 [cmd고객DB], [cmd닫기], [cmd등록]을 각각 클릭하여 'Click' 이벤트 프로시저를 작성합니다.



```
1 Private Sub cmd고객DB_Click()
2   Sheets("고객DB").Select
End Sub
```

- 1 'cmd고객DB' 명령 단추를 클릭했을 때 실행할 이벤트 프로시저를 시작합니다.
- 2 [고객DB] 시트를 선택합니다.



```
1 Private Sub cmd닫기_Click()
2   Unload Me
End Sub
```

- 1 'cmd닫기' 명령 단추를 클릭했을 때 실행할 이벤트 프로시저를 시작합니다.
- 2 현재 폼을 닫습니다.


```
1 Private Sub cmd등록_Click()
2   Unload Me
3   frm등록.Show vbModeless
End Sub
```

- ① 'cmd등록' 명령 단추를 클릭했을 때 실행할 이벤트 프로시저를 시작합니다.
- ② 현재 폼을 닫습니다.
- ③ [고객 등록] 폼을 모델리스로 열기합니다.


40 결과 확인하기 ① [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행]을 클릭합니다. 폼이 실행되면 ② [검색하기]를 클릭합니다. ③ 목록 상자에 고객 정보가 표시되면 한 고객을 클릭합니다. '고객상세정보' 프레임에 각 항목이 표시됩니다.



41 ① '고객상세정보' 프레임의 내용을 수정한 후 [수정 사항 등록]을 클릭합니다. 수정되었다는 메시지가 나타나면 ② [확인]을 클릭합니다. ③ 목록 상자에서 삭제할 고객을 클릭한 후 ④ [고객 삭제]를 클릭합니다. 삭제할 것인지 확인하는 메시지가 나타나면 ⑤ [예]를 클릭합니다.




42 폼 실행 프로시저 작성하기 ① [삽입]-[모듈]을 클릭합니다. ② [속성] 창에서 [이름]을 매뉴단추로 변경합니다.



실력향상 [메뉴단추] 모듈에는 [고객등록] 폼과 [고객 검색/삭제] 폼을 실행하는 프로시저와 [고객DB] 시트를 선택하고 프로그램을 종료하는 프로시저를 작성합니다. 작성된 프로시저는 [메뉴] 시트에서 단추(양식 컨트롤)에 연결합니다.

43 [메뉴단추] 모듈의 [코드] 창에 다음과 같이 네 개의 프로시저를 작성합니다.



Sub 고객등록()

① frm등록.Show vbModeless

End Sub

① [고객등록] 폼을 모달리스로 실행합니다.

Sub 고객검색()

① frm검색.Show vbModeless

End Sub

① [고객 검색/삭제] 폼을 모달리스로 실행합니다.

Sub 고객DB()

① Sheets("고객DB").Select

End Sub

① [고객DB] 시트를 선택합니다.

Sub 프로그램종료()

① If MsgBox("프로그램을 저장한 후 종료하시겠습니까?", vbYesNo) = vbYes Then

② ThisWorkbook.Save

③ Application.Quit

④ End If

End Sub



① 메시지 대화상자를 나타내고, [예]를 클릭하면 ② ~ ③이 실행되도록 하는 조건문입니다.

② 현재 통합 문서를 저장합니다.



③ 엑셀 프로그램을 종료합니다.

④ 조건문을 종료합니다.



44 단추에 프로시저 연결하기 ① [메뉴] 시트 탭을 클릭합니다. ② [개발 도구] 탭-[삽입] 그룹-[단추(양식 컨트롤)]을 클릭한 후 ③ 제목 아래 빈 공간에 드래그하여 삽입합니다. [매크로 지정] 대화상자의 ④ [매크로 이름]에서 [고객등록]을 클릭한 후 ⑤ [확인]을 클릭합니다.



45 ① 단추의 레이블을 **신규 고객 등록**으로 변경합니다. [신규 고객 등록] 단추를 **Ctrl** + **Shift** 를 누른 채 드래그하여 세 개를 복사합니다.



46 ① 두 번째 단추를 마우스 오른쪽 버튼으로 클릭한 후 ② [매크로 지정]을 클릭합니다. ③ [매크로 지정] 대화상자의 [매크로 이름]에서 ④ [고객검색]을 클릭한 후 ⑤ [확인]을 클릭합니다.



47 ❶ 단추의 레이블을 **고객 검색 및 수정**으로 변경합니다. ❷ 같은 방법으로 나머지 단추에도 매크로를 연결하고 레이블을 변경합니다.

